

### Seleção Múltipla

O comando de Seleção Múltipla geralmente é utilizado quando existe uma condição que possui uma única variável que pode assumir vários valores e de acordo com cada valor assumido é realizado uma tarefa.

Neste caso, apenas uma condição será satisfeita e a escolha da opção correta que satisfaça a valor da seleção, determina qual comando será executado.

#### Características:

- Presença de uma única variável assumindo vários valores, onde cada valor representa uma escolha de ação a ser desenvolvida pelo algoritmo.
- Variáveis sempre do tipo inteiro ou caractere.

Observar que dependendo da situação observada, uma das opções definidas pela escolha da opção correta depende do problema a ser solucionado. Assim, pode acontecer que não exista nenhuma opção disponível que satisfaça a escolha e desta forma o comando **Case Else** pode resolver essa situação.

#### Sintaxe:

```
Select Case <var>
  Case expre1 : <EXECUTA COMANDO 1>
  Case expre2 : <EXECUTA COMANDO 2>
  .....
Case Else :<EXECUTA OUTROS COMANDOS>
End Select { Finalizador}
```

Ou

```
Select Case <var>
  Case expre1
    <EXECUTA COMANDO 1>
  Case expre2
    <EXECUTA COMANDO 2>
  .....
Case Else
  <EXECUTA OUTROS COMANDOS>
End Select { Finalizador}
```

Onde:

<var> - expressão ou variável que normalmente é inteiro ou caractere

expre1, expre2 – são as opções a serem selecionadas pelo comando Select Case, que normalmente são possíveis valores assumidos pela variável <var>

Geralmente, é usado em menus ou mesmo para situações onde uma variável representa vários valores e para cada valor representado existe um desvio.

Considere a tabela abaixo onde são listados as formas de utilização do comando Select Case.

Opção Case	Descrição	Exemplos
<b>Is</b>	Uso com os seguintes operadores relacionais =, >, <, <>, >=, <=	<b>Case Is</b> = 5 <b>Case Is</b> = "Escola" <b>Case Is</b> > 5 <b>Case Is</b> <= 10
<b>To</b>	Uso para um conjunto de números os elementos	<b>Case</b> 5 <b>To</b> 10 <b>Case</b> 85 <b>To</b> 99 <b>Case</b> "A" <b>To</b> "D"
<b>Virgula</b>	Utilize a vírgula para incluir várias condições, para um único case	<b>Case</b> 1, 3, 9, 11 <b>Case</b> 1, <b>Is</b> >20, 10 <b>To</b> 15
<b>Nenhum operador</b>	O mesmo que "Is ="	<b>Case</b> 5 <b>Case</b> "Cidade"

Por exemplo 1: Considere que num problema onde se leia dois números e depois o sinal de operação (+, -, \* e /) e se deseja calcular a operação entre os dois elementos mostrando como saída o resultado final.

**Dados de entrada:**

A e B números inteiros

**Cálculo:**

De acordo com o valor do operador realiza-se uma operação

Se for + realiza a soma

Se for - realiza a subtração

Se for \* realiza a multiplicação

Se for / realiza a divisão.

**Saída do algoritmo:**

O resultado da operação entre a e b

Desta forma tem-se a resolução do problema utilizando-se comando IF encadeado

	B	C	D	E
1	B	Operador	Resultado	
2	2	+	O operador lido foi + A operação de 1+2 é igual a 3	
3	4	-	O operador lido foi - A operação de 3-4 é igual a -1	
4	0	/	não é possível realizar a operação	
5	4	/	O operador lido foi / A operação de 0/4 é igual a 0	
6	2	/	O operador lido foi / A operação de 4/2 é igual a 2	
7				

O código do exemplo acima:

```

Sub exemplo01()
'declaração de variaveis
Dim a As Double, b As Double, oper As String
Dim msg As String, r As Double
'leitura de dados
a = ActiveCell.Offset(0, -3).Value
b = ActiveCell.Offset(0, -2).Value
oper = ActiveCell.Offset(0, -1).Value
'Operação e calculos
If oper = "+" Then
    r = a + b
ElseIf oper = "-" Then
    r = a - b
ElseIf oper = "*" Then
    r = a * b
ElseIf oper = "/" Then
    If b <> 0 Then
        r = a / b
    End If ' quando for diferente de zero
End If
If oper = "/" And b = 0 Then
    msg = " não é possível realizar a operação"
Else
    msg = "O operador lido foi " & oper
    msg = msg & vbCrLf & " A operação de " & a & oper & b & " é igual a " & r
End If
ActiveCell.Value = msg
ActiveCell.Offset(1, 0).Select
End Sub

```

Observe que neste exemplo, usando o comando **If... Then... Else** encadeado, temos apenas teste para uma única variável que é **oper**, assumindo alguns valores que são os operadores soma(+), subtração(-), multiplicação(\*) e divisão(/).

Desta forma, para essa situação podemos utilizar o comando **Select Case** para resolver este problema de uma forma mais simples.

No exemplo abaixo, o comando **Select Case** vai possuir uma variável **op** que pode assumir vários valores do tipo caractere e para cada valor assumido ele realiza uma ação. Observe que esta estrutura de seleção múltipla é mais simples, pois os valores a serem escolhidos são definidos pelo próprio comando.

```

Select Case oper
Case "+"
    r = a + b
Case "-"
    r = a - b
Case "*"
    r = a * b
Case "/"
    If b <> 0 Then
        r = a / b
    End If ' quando for diferente de zero
End Select

```

O código completo será mostrado abaixo:

```

Sub exemplo02()
'declaração de variaveis
Dim a As Double, b As Double, oper As String
Dim msg As String, r As Double
'leitura de dados
a = ActiveCell.Offset(0, -3).Value
b = ActiveCell.Offset(0, -2).Value

```

```

oper = ActiveCell.Offset(0, -1).Value
'Operação e calculos
Select Case oper
Case "+"
    r = a + b
Case "-"
    r = a - b
Case "*"
    r = a * b
Case "/"
    If b <> 0 Then
        r = a / b
    End If ' quando for diferente de zero
End Select
If oper = "/" And b = 0 Then
    msg = " não é possivel realizar a operação"
Else
    msg = "O operador lido foi " & oper
    msg = msg & vbCrLf & " A operação de " & a & oper & b & " é igual a " & r
End If
ActiveCell.Value = msg
ActiveCell.Offset(1, 0).Select
End Sub

```

Por exemplo 1: seja dado como o status de um fornecedor de um determinado produto e realizado uma auditoria de tal forma que seja atribuído uma nota com base em alguns critérios de acordo com a tabela abaixo.

Nota a ser Atribuída	Mensagem a ser mostrada
nota < 30	DESQUALIFICADO
30 <= nota < 40	SITUAÇÃO CRITICA
40 <= nota < 50	ESTADO DE ALERTA
50 <= nota < 60	SOB MONITORAMENTO
60 <= nota < 70	ACEITÁVEL
70 <= nota < 80	BOM
80 <= nota < 90	MUITO BOM
Nota >=90	EXCELENTE

O objetivo é de acordo com a nota atribuída, mostrar a mensagem correspondente.

#### Dados de entrada:

Nota do tipo real

#### Cálculo:

Mostrar as mensagens de acordo com o valor lido para *nota*. Observe as condicionais a serem desenvolvidas.

#### Saída do programa:

A saída será uma das mensagens abaixo:

Mensagem a ser mostrada
DESQUALIFICADO
SITUAÇÃO CRITICA
ESTADO DE ALERTA
SOB MONITORAMENTO
ACEITÁVEL
BOM
MUITO BOM

EXCELENTE
-----------

Observe abaixo o novo código utilizando-se **Select Case**. Neste caso, como temos um conjunto de valores para cada escolha, precisamos utilizar a palavra **IS**.

```
Select Case nota
Case Is < 30 'aqui temos um conjunto de valores. Observe a palavra IS
s2 = "desqualificado"
Case Is < 40
s2 = "situação crítica"
Case Is < 50
s2 = "estado de alerta"
Case Is < 60
s2 = "sob monitoramento"
Case Is < 70
s2 = "aceitável"
Case Is < 80
s2 = "bom"
Case Is < 90
s2 = "muito bom"
Case Else
s2 = "excelente"
End Select
```

Observe que é possível inserir condições em cada opção escolhida usando comandos relacionais e uso da palavra **IS**.

#### Exercícios:

1. Elabore um programa em VBA que receba o salário atual de um funcionário e o código correspondente de um cargo do funcionário. Este código corresponde ao aumento que o funcionário terá em seu salário. O programa deverá mostrar o salário atual, a função do funcionário, o valor e percentual de aumento e o salário final.

CÓDIGO	CARGO	PERCENTUAL
1	Escriturário	40%
2	Digitador	35%
3	Analista	32%
4	Gerente	30%
5	Diretor	24%

#### Dados de entrada:

Sal do tipo real

Código do tipo inteiro (assume valores de 1 a 5)

#### Cálculo:

De acordo com o valor lido será realizado uma operação: (aqui não será validado o valor de entrada de dados)

Caso seja 1 mostra o cargo e calcula 40% sobre o salário

Caso seja 2 mostra o cargo e calcula 35% sobre o salário

Caso seja 3 mostra o cargo e calcula 32% sobre o salário

Caso seja 4 mostra o cargo e calcula 30% sobre o salário

Caso seja 5 mostra o cargo e calcula 24% sobre o salário

2. Faça um programa que receba um numero real e um outro contendo a escolha de 1 ate 4 e execute uma das operações listadas abaixo de acordo com uma entrada de dados a escolha do usuário. Se for digitada uma opção invalida, mostre uma mensagem de erro e termine a execução do programa. As opções de escolha para as operações são:

Numero	Operação
1	Elevar o numero ao quadrado
2	Elevar o numero ao cubo
3	Extrair a raiz quadrada
4	Extrair raiz cubica

**Dados de entrada:**

num do tipo real

op do tipo inteiro (assume valores de 1 a 4)

**Cálculo:**

De acordo com o valor lido será realizado uma operação: (aqui não será validado o valor de entrada de dados)

Caso seja 1 Elevar o numero ao quadrado

Caso seja 2 Elevar o numero ao cubo

Caso seja 3 Extrair a raiz quadrada

Caso seja 4 Extrair raiz cubica

3. Elabore um programa em VBA que receba a altura e o sexo de uma pessoa (M- masculino ou F – feminino) e mostre o seu sexo e o peso ideal, utilizando-se a formula abaixo:

M – para homens peso ideal =  $(72,7 * altura) - 58$

F – para mulheres peso ideal =  $(62,1 * altura) - 44,7$

**Dados de entrada:**

altura do tipo real

sexo da pessoa do tipo string (assume valores M ou F, por exemplo)

**Cálculo:**

De acordo com o valor lido será realizado uma operação: (aqui não será validado o valor de entrada de dados)

Caso seja M calcular o peso ideal para Homens

peso ideal =  $(72,7 * altura) - 58$

Caso seja F calcular o peso ideal para Mulheres

peso ideal =  $(62,1 * altura) - 44,7$

4. Elabore um programa em VBA que receba a quantidade de um determinado produto e o seu codigo de procedencia. Assim, de acordo com esse codigo mostre a procedencia e o valor total a pagar de acordo com a tabela abaixo:

CÓDIGO	PROCEDÊNCIA	PREÇO POR UNIDADE
1	Sul	R\$ 12,45
2	Norte	R\$ 21,45
3 ou 4	Leste	R\$ 23,45
5 ou 6	Oeste	R\$ 25,34
7 ate 10	Sudeste	R\$ 28,56
11 até 20	Nordeste	R\$ 21,87
21 ou 22	Sudoeste	R\$ 28,65

23 ate 50	Noroeste	R\$ 29,43
-----------	----------	-----------

O valor a pagar é a multiplicação da quantidade pelo valor de cada unidade.

**Dados de entrada:**

Quantidade do tipo inteiro

Código de origem que é do tipo inteiro

**Cálculo:**

De acordo com o valor lido será realizado uma operação: (aqui não será validado o valor de entrada de dados)

5. Elabore um programa em VBA que receba o numero de horas trabalhadas, o valor do salario minimo e o código (1- Escritório, 2- Produção, 3-Portaria, 4-Gerência) referente a área onde o funcionário trabalha. Calcule e mostre o salario final a receber de acordo com as seguinte regras:

- a. O valor da hora trabalhada será obtida de acordo com a tabela abaixo:

CÓDIGO	CARGO	VALOR	IMPOSTO
1	Escritório	R\$ 59,90	5%
2	Produção	R\$ 89,90	6%
3	Portaria	R\$ 120,98	8%
4	Gerência	R\$ 130,87	10%

- b. Quando o cargo for de Gerência e o valor da hora estiver inferior a  $\frac{1}{4}$  do salario minimo, será utilizado este novo valor. Para os cargos de Escritorio, Portaria e Produção se o valor da hora estiver abaixo de  $\frac{1}{8}$  do salario minimo será utilizado o novo valor acrescido de 5% no valor da hora trabalhada.
- c. O salário bruto será a quantida de horas trabalhadas multiplicado pelo valor da hora trabalhada.
- d. Quando o Cargo for do Escritório ou Portaria e o salário bruto estiver entre R\$ 2.300,00 e R\$5.000,00 será dado um acrescimo de 13,28% no salario bruto.
- e. O imposto será calculado como um percentual obtido na tabela do item a) e será aplicado ao salário bruto.
- f. O salário final será o salário bruto menos o valor do imposto.
- g. Mostre todos os dados