

Aula 04 – Estruturas de Seleção

Introdução


A estrutura de seleção permite que se decida executar determinado comando ou não, dependendo de uma determinada condição ser satisfeita ou não. A seleção é a possibilidade de escolha de execução entre, por exemplo dois comandos que é realizada quando esta estrutura avalia uma condição que assume dois valores lógicos: Verdadeiro (*true*) ou Falso (*false*)

Uma das **características desta estrutura de controle é a presença de condição**, ou seja, a presença de uma expressão lógica que tem como resultado o valor lógico VERDADEIRO ou FALSO, dependendo do resultado de uma operação lógica, valor lógico ou comparação.

Outra característica importante deste tipo de estrutura de controle é que este comando realiza o **desvio no fluxo de execução** de um programa.

O que **significa um desvio no fluxo de execução**? Normalmente durante a execução de um programa ou algoritmo o fluxo de execução é realizada de forma sequencial, ou seja, um comando de cada vez em sequencia. Desta forma, uma estrutura de seleção permite realizar uma pergunta cuja resposta pode ser Verdadeiro ou Falso e para cada resposta dada ele executar um comando diferente.

Seja por exemplo perguntar se um dado numero é par ou impar? A resposta neste caso, depende da forma com que a pergunta é realizada. Considere o fluxograma abaixo:



Neste fluxograma a pergunta que se faz é se $n \bmod 2 = 0$? Se a resposta a essa pergunta for **VERDADEIRA** então ele deve mostrar que o numero escolhido é PAR, caso contrário, a resposta será **FALSO** e ele deve mostrar que o numero escolhido é IMPAR.

Assim, observamos que existe um desvio no fluxo de execução através da condição executada que permite a possibilidade de realizar um comando ou um conjunto de comandos (quando existem blocos de execução).

Como foi visto acima, uma condição é uma pergunta cuja resposta pode ser VERDADEIRO ou FALSO. Desta forma, uma condição pode ser realizada através de duas formas:

- Através de uma **expressão lógica** utilizando operadores relacionais ($>$, $>=$, $<=$, $<$, $=$, $<>$) ou através de **operadores lógicos** (AND, OR, NOT)

Por exemplo:

Se A e B são duas variáveis inteiras e $A = 10$ e $B = 20$ tem-se os seguintes valores lógicos, dependendo do operador relacional escolhido:

- $(A=B)$ condição é FALSO

- (A = A+ 1) condição é FALSO
- (B = A +10) condição é VERDADEIRO
- (A=B OR B > 3) condição é VERDADEIRO

b) Através de **variáveis lógicas**, ou seja, variáveis que tem como o tipo de dado chamado de BOOLEAN na linguagem VBA. Este tipo de variável assume como valores somente TRUE ou FALSE. Além disso, podem ser atribuídos operações com operadores lógicos ou relacionais

Por exemplo:

Se L é uma variável do tipo **BOOLEAN** então L pode assumir valor **TRUE** ou **FALSE**. Assim, o comando L= TRUE- fará com que seja atribuído o valor VERDADEIRO à variável L.

A seguir vamos estudar as estruturas de controle utilizadas pela linguagem VBA

A estrutura condicional pode ser simples ou composta, dependendo do desvio a ser utilizado, ou seja, dependendo do caminho que a condição pode demandar.

As estruturas de controle determinam o curso de ações de um algoritmo ou programa. A lógica de programação flui através das instruções da esquerda para a direita e de cima para baixo. As instruções de controle, ou seja, os comandos que controlam a tomada de decisões e as iterações podem alterar a ordem de execução das instruções.

1- Seleção simples

Neste caso, é realizado somente um desvio no fluxo de execução, de tal forma que é executado um comando ou conjunto de comandos. Este desvio será realizado quando a CONDIÇÃO for **VERDADEIRO** e neste caso caso, não existe a instrução SENÃO, ou seja, ele não executa nada quando a CONDIÇÃO for **FALSO**.

Tem-se abaixo a sintaxe do comando

SINTAXE:

```
If <condição> Then
    <Conjunto de comandos>
End If
```

Este comando testa uma única condição. Observe que isso significa a possibilidade de se ter vários operadores de comparação ou lógicos, que resultam em somente um valor lógico **VERDADEIRO** ou **FALSO**.

Exemplo:

Considere obter a leitura de dois números inteiros e efetuar a soma destes dois elementos. O resultado é mostrar somente quando essa soma for maior do que 5.

Assim temos que realizar a leitura de dois números, depois realizar a soma entre eles e armazenar o resultado em uma variável. O próximo passo é testar se essa soma é maior que 5.

Observe que se a condição não for verificada (não for VERDADEIRO) o programa não mostra nenhuma mensagem.

Considere a planilha de dados abaixo:

	A	B	C	D	E
1	num1	num2	Soma	Mensagem	
2	2	3	5	A soma é Valida	
3	1	2	3		
4	-2	3	1		
5	4	7	11	A soma é Valida	
6					
7					

O código do exemplo pode ser visto abaixo:

```

Sub Exemplo01 ()
'declaração de variaves
Dim num1 As Integer, num2 As Integer
Dim soma As Integer, msg As String
'Entrada de dados. Obter dois numeros
num1 = ActiveCell.Offset(0, -2).Value
num2 = ActiveCell.Offset(0, -1).Value
'Operações / Calculos
soma = num1 + num2
If (soma >= 5) Then
    msg = " A soma é Valida "
End If
'Mostrar Resultados
ActiveCell.Value = soma
ActiveCell.Offset(0, 1).Value = msg
ActiveCell.Offset(1, 0).Select
End Sub

```

2 - Seleção composta

Neste caso, é realizado somente um desvio no fluxo de execução de tal forma que possa ser executado um comando quando a condição for **VERDADEIRO** e outro comando quando a condição for **FALSO**. Observa-se que em qualquer situação o desvio do fluxo de execução do programa irá acontecer, quer seja no **VERDADEIRO** ou no **FALSO**.

Tem-se abaixo, a sintaxe do comando

SINTAXE:

```

If <condição> Then
    <Conjunto de comandos 1>
Else
    <Conjunto de comandos 2>
End If

```

Exemplo:

O procedimento a seguir efetua o cálculo da média aritmética da leitura de duas notas de um aluno e avalia a situação quanto à aprovação. A situação é que se a nota for maior ou igual a 6,0 o aluno estará "APROVADO" e caso contrário, o aluno estará "REPROVADO".

Considere a planilha de dados?

	A	B	C	D	E
1	num1	num2	Media	Situação	
2	6,8	7,8	7,3	O aluno esta APROVADO	
3	3,5	5,2	4,4	O alunos esta REPROVADO	
4	3,9	8,2	6,1	O aluno esta APROVADO	
5	4,2	3,1	3,7	O alunos esta REPROVADO	
6					
7					

O código do exemplo pode ser visto abaixo:

```
Sub exemplo02()
'Declaração de variaveis
Dim n1 As Double, n2 As Double
Dim media As Double, msg As String
'Entrada de dados
n1 = ActiveCell.Offset(0, -2).Value
n2 = ActiveCell.Offset(0, -1).Value

'Operações / Calculos
media = (n1 + n2) / 2
If (media >= 6) Then
    msg = " O aluno esta APROVADO "
Else
    msg = " O alunos esta REPROVADO "
End If
'Mostrar resultado
ActiveCell.Value = media
ActiveCell.Offset(0, 1).Value = msg
ActiveCell.Offset(1, 0).Select
End Sub
```

Observe a seguinte execução:

Entrada de dados	resultado
Suponha que os dados digitados sejam n1=3,5 e n2 = 4,0	O alunos esta REPROVADO
Suponha que os dados digitados sejam n1=7,5 e n2 = 5,0. A resposta será:	O aluno esta APROVADO

Observa-se que em qualquer situação uma das mensagens será mostrada para o usuário.

3 - Seleção composta encadeada

Neste caso, temos a possibilidade de verificar **mais de uma condição** de tal forma que se caracterize como um encadeamento de comandos condicionais contendo várias condições e executando um dos vários blocos de instruções.

Independentemente da quantidade de condições a ser verificado, normalmente o resultado do desvio no fluxo de execução será único. Isso significa que o resultado final de todos os comandos condicionais é sempre um resultado ou uma mensagem.

Tem-se abaixo a sintaxe do comando

SINTAXE:

```
If <condição 1> Then
    <Conjunto de comandos 1>
ElseIf <Condição 2> Then
    <Conjunto de comandos 2>
Else
    <Conjunto de comandos 3>
End If
```

Observe que existem vários comandos condicionais **If's** e vários **ElseIf's** e todos os comandos terminam com um único **End If**.

Exemplo:

O procedimento a seguir efetua o cálculo da média aritmética da leitura de duas notas de um aluno e avalia a situação quanto à aprovação. A situação é que se a nota for maior ou igual a 6,0 o aluno estará "APROVADO". Se a nota estiver entre 4,0 e 6,0, o aluno estará em "RECUPERAÇÃO" e caso contrário, o aluno estará "REPROVADO".

Considere a planilha de dados:

	A	B	C	D
1	num1	num2	Media	Situação
2	6,8	7,8	7,3	O aluno esta APROVADO
3	3,5	4,0	3,8	Aluno esta REPROVADO
4	3,9	8,2	6,1	O aluno esta APROVADO
5	4,2	3,1	3,7	Aluno esta REPROVADO
6	3,5	7,0	5,3	Aluno esta em RECUPERAÇÃO
7	7,5	7,0	7,3	O aluno esta APROVADO

Conforme a entrada de dados tem-se os seguintes resultados:

Entrada de dados	resultado
Suponha que os dados digitados sejam n1=3,5 e n2 = 4,0	Aluno esta REPROVADO
Suponha que os dados digitados sejam n1=3,5 e n2 = 7,0,0	Aluno esta em RECUPERAÇÃO
Suponha que os dados digitados sejam n1=7,5 e n2 = 7,0,0	O aluno esta APROVADO

Assim o programa em VBA que resolve o nosso problema:

```
Sub exemplo03()
'Declaração de variaveis
Dim n1 As Double, n2 As Double
Dim media As Double, msg As String
'Entrada de dados
```

```
n1 = ActiveCell.Offset(0, -2).Value  
n2 = ActiveCell.Offset(0, -1).Value
```

```
'Operações / Calculos
```

```
media = (n1 + n2) / 2
```

```
If (media >= 0 And media < 4) Then
```

```
    msg = "Aluno esta REPROVADO "
```

```
    ElseIf (media >= 4 And media < 6) Then
```

```
        msg = " Aluno esta em RECUPERAÇÃO "
```

```
    Else
```

```
        msg = " O aluno esta APROVADO "
```

```
End If
```

```
'Mostrar resultado
```

```
ActiveCell.Value = media
```

```
ActiveCell.Offset(0, 1).Value = msg
```

```
ActiveCell.Offset(1, 0).Select
```

```
End Sub
```

Observação:

- a) O comando termina com apenas um **End IF**.
- b) A nova condição é testada ou verificada utilizando-se o comando **ElseIf**.
- c) O ultimo comando **Else** , refere-se ao ultimo comando **ElseIf**.

Existe outro comando que é o de Seleção Múltipla que será visto em outra aula.

Exercícios:

1. Sejam dados o peso e altura de uma pessoa, elabore um programa que calcule o IMC (Índice de Massa Corporea) e mostre a situação de acordo com a tabela abaixo:

Resultado	Situação
Abaixo de 17	Muito abaixo do <i>peso</i>
Entre 17 e 18,49	Abaixo do <i>peso</i>
Entre 18,5 e 24,99	<i>Peso normal</i>
Entre 25 e 29,99	Acima do <i>peso</i>
Entre 30 e 34,99	<i>Obesidade I</i>
Entre 35 e 39,99	<i>Obesidade II (severa)</i>
Acima de 40	<i>Obesidade III (mórbida)</i>

Dados de Entrada:

Altura que é um numero real

Peso que é um numero real

Operação a ser realizada:

$$\text{IMC} = \text{peso} / (\text{altura})^2$$

Saida do programa:

A saída do programa é uma das 7 mensagens mostrada no quadro acima, sendo que ela depende do valor do IMC.

2. Elabore um programa que entre com os valores de A, B e C que correspondam aos valores de $Ax^2 + Bx + C = 0$, que representa uma equação do 2º grau. O objetivo do programa é mostrar quando possível as raízes X_1 e X_2 . Um esboço do algoritmo obtido do livro da Ascêncio pode ser visto abaixo:

ALGORITMO

```
    DECLARE a, b, c, delta, x1, x2 NUMÉRICO
    LEIA a, b, c
    SE a = 0
    ENTÃO ESCRVA "Estes valores não formam uma equação de segundo grau"
    SENÃO INÍCIO
        delta ← (b * b) - ( 4 * a * c)
        SE delta < 0
            ENTÃO ESCRVA "Não existe raiz real"
        SE delta = 0
            ENTÃO INÍCIO
                ESCRVA "Existe uma raiz real"
                x1 ← (-b) / ( 2 * a)
                ESCRVA x1
                FIM
        SE delta > 0
            ENTÃO INÍCIO
                ESCRVA "Existem duas raízes reais"
                x1 ← (- b +  $\sqrt{\Delta}$ ) / ( 2 * a)
                x2 ← (- b -  $\sqrt{\Delta}$ ) / ( 2 * a)
                ESCRVA x1, x2
                FIM
            FIM
    FIM
FIM_ALGORITMO.
```

3. Elabore um programa que entra com dois números inteiros e mostre todas as relações existentes entre eles, utilizando-se operadores relacionais.

Dados de Entrada:

Num1 – numero inteiro

Num2 – numero inteiro

Operação a ser realizada:

Não existe nenhuma operação a ser realizada, apenas mensagens a serem mostradas de acordo com as condições entre dois dados.

Saída do programa:

A saída será uma das mensagens abaixo:

>	Um numero maior que outro
>=	Um numero maior ou igual a outro
=	Um numero igual a outro
<	Um numero menor que outro
<=	Um numero menor ou igual a outro
<>	Um numero diferente que outro

A saída do programa é uma das 7 mensagens mostrada no quadro acima, sendo que ela depende dos valores relativos a cada valor de operador relacional.

4. Faça um programa que entre com 3 números e mostre-os em ordem crescente. Suponha que os números lidos pelos usuários sejam diferentes. Olhe o algoritmo do exercício resolvido 04 na página 62 do livro da Ascêncio.
5. Faça um programa que receba 04 valores A, B, C e D, números inteiros. Os valores permitidos de D são 0, 1 e 2. Quando o valor de D for igual a zero (0) a saída será os números em ordem crescente. Quando o valor de D for igual a um (1) a saída será os números em ordem decrescente. Quando o valor de D for igual a dois(2) a saída será que o menor numero deverá ficar entre os outros dois números. *Olhe o algoritmo do exercício resolvido 07 na página 63 do livro da Ascêncio.*